

MODEL VALIDATION AND TESTING OF INTERNATIONAL SPACE STATION STRUCTURES USING MSC/NASTRAN

By

Tarun Ghosh
Jeyaraj Wilson
John Chang
Boeing North American, Inc.
Canoga Park, California

ABSTRACT

In order to perform analysis to support the design of the International Space Station, mathematical models of the components were generated in the form of finite element models. Test verification of these models is required before the components can be certified for launch. Modal testing is one of the several tests performed on the components. The purpose of the modal tests is to show that the mathematical models have the same dynamic characteristics as the physical models. Namely, they have the same frequencies, modeshapes and mass properties. As part of pre-test analysis, fixtures and flexures simulating the boundary condition need to be validated. When the test article differs from the flight hardware, it too has to be qualified. Besides, shaker locations for application of forces and accelerometer locations for the collection of data need to be identified. It is important to identify the target modes. As part of post-test analysis, the test modeshapes are used to validate the mathematical models. For better correlation with test data, the mathematical model may need to be updated. The purpose of this paper is to show how efficiently MSC/NASTRAN can be used for model validation and model updating.

INTRODUCTION

In today's world, the design and construction of large structures often requires the combined effort of several organizations. The International Space Station (ISS) is one such structure and is shown in Figure-1. It is an international joint venture and countries across the globe are participating. It is a giant station that will operate in space, and mankind will use it for research, commercial production and voyage to outer space. When completed, it will be about 350 feet long and weigh about a million pounds.

Mathematical analysis is an indispensable tool in the design of large structures such as the ISS. To do this, mathematical models of the different components are required. The mathematical models are built using finite element (FE) techniques and are therefore called FE models. The technique involves representing a component as assemblage of elements, and is described in Craig¹, Craig² and Cook, et al³. The dynamic response of the structures in the form of loads, displacements and stresses at the boundary and interior of components is obtained using a modal superposition technique, the details of which can be found in Craig¹, Craig² and Cook, et al³. So it is necessary that the mathematical modeshapes correlate with the test modeshapes and have the same frequencies.

This paper addresses the structural aspect of modal testing. Though the mathematics involved are fairly simple, there are different aspects that the structural engineer should be aware of. How does the test article compare with the flight (real) hardware? Does the test fixture/flexure properly simulate the flight boundary condition? What should be the target modes and where should one put the accelerometers and shakers? As part of post-test analysis, the mathematical model has to be updated so that the modeshapes and frequencies correlate with those from the test. This can be a time consuming pains-taking task.

There is often a tendency to overdo modal testing. There are many modal testing packages that try to convince the structural engineers that modal testing without using these packages would be a total failure. Though modal testing, model correlation and model updating of certain components can be a real challenge, there is no single technique or package that will work like a charm. Besides, as these packages are like black boxes, they should be used with caution. The MSC/NASTRAN library provides software in the form of Direct Matrix Abstraction Programs (DMAPs) for modal testing. These DMAPs can be easily modified to handle the challenge. That is exactly what was done in the modal testing of a fairly complex part of the ISS known as the P6 Cargo Element, which is shown in Figure 2. It is a component that weighs about 35,000 lbs. and is about 30 ft long. What makes the task complicated is that some parts are flexible while others are stiff.

THEORY

In finite element analysis, a physical model of a structure with applied loads is represented mathematically as

$$[m]\{\ddot{u}\} + [c]\{\dot{u}\} + [k]\{u\} = \{f(t)\} \quad (1)$$

$[m]$, $[c]$ and $[k]$ are the mass, stiffness and damping matrices, respectively. $\{u\}$, $\{\dot{u}\}$, and $\{\ddot{u}\}$ represent the generalized displacements, velocities and accelerations at the physical degrees of freedom and $\{f(t)\}$ represents the generalized forces at the physical degrees of freedom. The equations are generated by assembling elements defined mathematically in the physical domain.

For linear elastic structures, the equations in (1) can be transformed to the modal domain and expressed as follows

$$[M]\{\mathbf{h}\} + [C]\{\dot{\mathbf{h}}\} + [K]\{\mathbf{h}\} = \{F(t)\} \quad (2)$$

where $[M]$, $[C]$ and $[K]$ are the system modal mass, damping, and stiffness matrices. $\{\mathbf{h}\}$ is the generalized (modal) displacement vector and $\{F(t)\}$ is the modal load vector. The method of generating the equations in (2) from the equations in (1) requires an eigensolution of the undamped free system. That is, in equations (1), $[c]$ is assumed to be a null matrix and $\{f(t)\}$ is assumed to be a null vector. The eigenvectors so obtained are used to generate an eigenvector matrix $[\Phi]$. For linear systems,

$$\begin{aligned}\{u\} &= [\Phi]\{\eta\} \\ \{\dot{u}\} &= [\Phi]\{\dot{\eta}\} \\ \{\ddot{u}\} &= [\Phi]\{\ddot{\eta}\}\end{aligned}\tag{3}$$

Then, by replacing $\{u\}$, $\{\dot{u}\}$ and $\{\ddot{u}\}$ in (1) by the right hand sides of (3) and pre-multiplying both sides by $[\Phi]^T$, one obtains (2). Actually, $[C]$ in (2) comes from test data or other sources. For details of the method, one can refer to Cook³ and Meirovitch⁶.

From the above, the importance of the structural modeshapes and their associated frequencies can be easily seen. For the analytical solution from the mode-superposition method to be trusted, the test modeshapes and their frequencies must agree with those from analytical model of equation (1). Frequency being a scalar quantity, its agreement is based on its numerical value. But modeshapes being vectors, their agreement is based on correlation functions.

There are two correlation functions widely used. One is the modal assurance criteria called MAC based on the dot product of vectors. The other is the orthogonality of the modeshapes with respect to the mass matrix to be called ORTHO in this paper. In either case, the aim is to get an identity matrix. While for a lumped mass model, it is obvious that MAC should be $[I]$, for a coupled mass model, MAC needs to be approximately $[I]$ because when there are many dofs, lumped mass and coupled mass should yield approximately the same kind of results.

What follows next are the different steps of the structural aspects of the modal testing and the model updating and verification procedure.

Selection of Target Modes:

The flight model of the P6 Cargo Element has 518 vibration modes below 70 Hz. So a criteria was required to select the dominant modes. Dominant modes were selected based on the effective mass of P6 vibration modes and the contribution of individual vibration modes to interface forces at the P6/orbiter and internal component interface forces.

From (2), one would expect the terms of $[m][\Phi_i]$ to be large for modes that are dominant. So if the modal effective mass, which is the summation of the terms of $[m][\Phi_i]$ associated with any direction, is significant, it would indicate the mode is dominant. DMAPs that perform this function in MSC/NASTRAN are readily available. Modes that have large effective translation mass are selected as target modes.

From modal synthesis, the response $\{s\}$ of the data recovery items is given by

$$\{s\} = [T_1]\{u_c\} + [T_2]\{\ddot{u}_c\}$$

where $[T_1]$ and $[T_2]$ are the displacement and acceleration data recovery matrices computed by standard MSC/NASTRAN DMAPs and $\{u_c\}$ and $\{\ddot{u}_c\}$ are the displacements and accelerations of the vibration and boundary degrees of freedom of the component punched out using DISP and ACCE case control commands. The computation was performed using the code named 'dommod2' developed as part of this procedure. Modes that have large contribution or large effective mass were selected as target modes.

Qualification of Fixture and Flexures:

While the flight hardware will be mounted on the orbiter, the test article will be mounted on the fixture. It is essential that the fixture be stiff. The flexures will simulate the orbiter boundary

condition. The qualification method used was to compute the frequencies and modeshapes of the flight model once using orbiter boundary condition and once using fixture/flexures. The frequencies of the target modes differed by less than 2%. The ORTHO check showed, for dominant modes, 0.90 or higher in diagonal terms and 0.10 or less in off diagonal terms. Of the nine target modes, only the highest two (modes 19 and 66) had more than 10% coupling with the closest (non-target) mode. Based on the results, it is concluded that the flexures correctly represent the orbiter boundary condition, and that the test setup is acceptable for identifying the ten target modes. In the ORTHO check the mass matrix used was that of the P6 Cargo Element without fixtures and flexures. This was a two step process. In the first step, the modes were computed when using fixtures and flexures and saved. The fixtures and flexures were represented as superelements and input using CSUPER entry. In the second run, the orbiter boundary condition and modes from first run were used. The P6 Cargo Element mass matrix was used in the second run.

Qualification of the Test Article:

The test article of the P6 Cargo Element was somewhat different from the flight hardware. While the flight hardware has three stowed radiators, the test article has just one stowed radiator that is strapped down with foam placed between radiator panels. The qualification was performed in two steps. In the first step interface forces at the P6/orbiter and internal component interfaces were computed once using lumped mass radiators and once using detailed radiators. The results were close. This showed that a lumped mass representation of the radiator did not affect the primary target modes. In order to determine if the elimination of the two radiators would give system modeshapes comparable to those of the lumped mass representation (and detailed representation), an orthogonality check was performed. The test showed that the primary target mode frequencies may shift, but the modeshapes were primarily unaffected.

Selection of Accelerometer Locations:

In all over 400 tri-axial accelerometers were used of which 38 were on the fixture and the rest were on the test article (P6). While the fixture accelerometer locations were fixed (based on previous tests), selection of P6 Cargo Element accelerometer locations was part of pre-test analysis. The P6 Cargo Element accelerometer locations were selected so that the Guyan reduced model modeshapes based on accelerometer locations would agree with those of the unreduced model (with orbiter boundary condition) as measured by the numerical values of frequencies and ORTHO and MAC checks (using the Guyan reduced mass matrix). Since accelerometers may be placed at any of the thousands of accessible grid points, several trials were required. Alternatively, Guyan reduced modes expanded to the full G-set along with the unreduced mass matrix and modeshapes could have been used. But the former method was found to be superior.

Selection of Shaker Locations:

One of the tasks of the P6 Cargo Element modal pre-test analysis was to identify the shaker locations. In the language of the Finite Element (FE) model, the task was to identify the dofs where a force excitation would excite the target modes. The aim was to have a minimum number of shaker locations (dofs). The locations were to be accessible and selected such that applying a force of 50-100 lbs. would not damage the part. 13 grids representing the four Integrated Equipment Assembly (IEA) free end corner grids, four longspacer free end corner grids, four IEA/longspacer interface corner grids and a Beta Gimbal Assembly (BGA) grid were selected. Using MSC/NASTRAN frequency response function (FRF) computation in Solution 111, the FRF of displacement to force at each of the translation dofs of the 13 grids were plotted for damping of 0.5% and 2.0%. The (local) peaks of the plots identified the modes that the shaker at that dof would excite. Eventually five degrees of freedom were selected for force application that would excite all the target modes.

Visualization of Vibration Modes:

Plotting of the vibration modes is a simple yet useful step. Plotting of the A-set analytical modes is fairly easy. First the displacements of the A-set dofs were output in punch form using a DMAP. A computer code was used to generate a universal file with modeshapes that could be plotted in IDEAS. Plotting of the test modes was even simpler. The universal file with test modeshapes generated by LMS software needed some editing before they could be plotted in IDEAS.

Post Processing and Model Updating:

Post processing of the data involves identifying the bad accelerometer measurements and updating the mathematical model. The task of identification is a judgment call based on the expected behavior of the mathematical model which comes from experience with the hardware. After the elimination of the bad accelerometers, measurements from 392 tri-axials were left of which 38 were on the fixture. There is no cookbook methodology for model updating. It is highly dependent on the individual engineer. Knowledge of the hardware/model is essential. One has to identify the locations of large deflection to improve orthogonality. Often this is done based on measured deflection or strain energy of selected grids. A FORTRAN code that had such useful features was developed and called ORTHO4. It could also perform a part MAC check and identify dominant degrees of freedom in any term of the ORTHO matrix. Note that for model validation/updating the orthogonality check was performed in the A-set.

Use of Test Modes in Modal Synthesis:

At times the reason for modal test is to meet certification requirement that the flight hardware does not impose excessive loads on the shuttle. Then one would like to compute the responses obtained using test modes. But as the test does not provide a mass matrix or full G-set modeshapes, an approximate method may be used. In this method the test modes are expanded to the G-set using the A-set to G-set transformation for the analytical model. The expanded G-set modes are used to generate the Craig-Bampton reduced model and matrices. The method is

approximate because the expanded G-set modes are not orthogonal with respect to the analytical G-set mass matrix.

Use of Guyan Reduced Fixture/Flexure Model:

Since the fixture has some flexibility, ORTHO check using fixture/flexure deflections along with test article deflections may be desired. Again, since the fixture/flexures is represented in Craig-Bampton reduced form that has only boundary and vibration dofs, the fixture/flexures model must be Guyan reduced (to the accelerometer dofs) and expanded to the G-set before Craig-Bampton reduction.

Software:

Because the orthogonality check is a very useful feature that is used for a variety of needs, it is important that any procedure for the orthogonality check be flexible. When one is performing an orthogonality check using large matrices such as the G-set mass matrix when doing model validation in the full G-set as part of pre-test analysis, it is convenient to perform this in the MSC/NASTRAN environment. DMAP 'pchdispa' (in MSC/NASTRAN library) or something similar may be obtained from the MSC/NASTRAN library to output displacement modeshapes in DMIG form generated in Solution 103 for model-1. Then the DMIG displacements can be input in the second Solution 103 run and the mass matrix of model-2 may be used to do the ORTHO or MAC check using a DMAP such as 'premaca' from MSC/NASTRAN library or something similar. But when performing the orthogonality check in the reduced (A-set) domain as in model updating in post-test analysis, the matrices are much smaller and one may do so outside of MSC/NASTRAN using a software such as FORTRAN code ORTHO4 developed as part of this procedure. ORTHO4 has such features as a part MAC check and a user provided selection of modes.

Codes to translate modeshapes from DMIG form to universal file form or do the reverse were generated. The former was needed for plotting while the latter was needed for use in the orthogonality check using MSC/NASTRAN or ORTHO4.

A code to determine modal participation and perform data recovery from the output transformation matrix and component boundary and vibration degree of freedom was named DOMMOD2 and was heavily used.

Figures 3 through 6 give several DMAPs in Version 68.1 of MSC/NASTRAN. The DMAP of Figure-3 is used to output G-set modeshapes in DMIG form from Solution 103. This or a similar one can be found in MSC/NASTRAN library. The DMAP of Figure-4 is a modification of the one in the MSC/NASTRAN library. This one can output ORTHO and MAC matrices, as well as output the G-set mass, and modeshape matrices. The DMAP of Figure-5 is to output the A-set modeshape from Solution 103. The DMAP of Figure-6 is to output the U-set modeshapes from Solution 103. The last two DMAPs will also output in punch form the eigenvalue table.

ACKNOWLEDGMENT

This work was performed at Rocketdyne Division, Boeing North American, Inc. The material presented here is based on work performed for the International Space Station under NASA contract HX3210. The authors wish to acknowledge the support of Rocketdyne and NASA management. The authors also wish to thank Massoud Nikki, Ray Shirr and Wayne Smith for their technical support.

REFERENCES

1. Craig, R. R., "A Review of Time-Domain and Frequency Domain Component Mode Synthesis Methods," Int. J. Analytical and Experimental Modal Analysis, Vol. 2, No. 2, 1987.
2. Craig, R. R., "Structural Dynamics", John Wiley & Sons, New York, 1981.
3. Cook, R. D., Malkus, D. S. and Plesha, M. E., "Concepts and Applications of Finite Element Analysis", John Wiley & Sons, 3rd Edition, 1989.
4. Meirovitch, L., "Elements of Vibration Analysis", Second Edition, McGraw-Hill Book Co, New York, 1986.
5. Ghosh, T. K., "Improved Method of Modal Synthesis in the analysis of International Space-station Structures", 38th AIAA/ASME/ASCE/AHS/ASC Structural Dynamics, and Materials Conference, Proceedings, Kissimmee, Florida, April 7-10, 1997.
6. Blakely, K. And Rose, T., "Cross-Orthogonality Calculations for Pre-Test Planning and Model Validation", Proceedings MSC World user's Conference, 1993.




FIGURE 1: INTERNATIONAL SPACE STATION




FIGURE 2: P6 CARGO ELEMENT

```

compile sedrcvr, souin=mscsou, noref
alter 50
matgen eqexins/intext/9/0/lusets $
mpyad intext,ug,/ugvext/1 $
matmod ugvext,eqexins,,,/mat1,/16/0 $ CH1997
OUTPUT2 MAT1,,,,/-1/11// $ AD1997
OUTPUT2 UGVEXT,EQEXINS,,,,/-1/12// $ AD1997

```

FIGURE-3: DMAP FOR G-SET MODESHAPE OUTPUT IN SOL 103 IN VERSION 68.1

```

$
$   tested on system built on 5-31-94
$
$   filename - premaca.v68
$
$   Cross-Orthogonality calculation DMAP for SOL 103 - V67
$
$   Reads DMIG input modes and compares to calculated modes
$   Input modes are from full (or larger) run
$
$   Input:      UGVEXT as DMIG (input modes)
$
$   Outputs:    PHIEXT as DMIG (computed modes)
$               PHIEX2 as DMIG (system modes without reduction)
$               MAAEXT as DMIG (ASET mass)
$               (all are for subsequent post-test run)
$
$   Computes:   MAC (modal assurance criterion)
$               CHECKIT (UGVEXT(T)*MAA*UGVEXT)
$               ORTHOA (PHIA(T)*MAA*UGVEXT)
$
$   COMPILE MODERS, SOUIN=MSCSOU, LIST, REF $
$   ALTER 5
$   TYPE PARM,,I,N,EVAL=-1,MODENO=0 $
$   TYPE PARM,,I,N,MODNOM1,MOLEFT $
$   TYPE PARM,NDDL,I,N,LUSETS,ZUZR1 $
$   TYPE DB,MATPOOL $
$   TYPE DB,ZUZR01 $
$   $ALTER 86 $ V66
$   $ALTER 89 $ V66A AND V66B
$   ALTER 116 $ BEFORE RETURN
$
$   MODAL ASSURANCE CRITERIA
$
$       MTRXIN ,,MATPOOL,EQEXINS,SILS,/UGVEXT,,/LUSETS/S,N,EVAL $ DE1997
$       INPUTT4 /UGVEXT,,,,/1/12/-1/-1 $ AD1997
$       EVAL=1 $ AD1997
$   IF (EVAL=1) THEN $
$       MESSAGE // ' ' $
$       MESSAGE //'MODAL ASSURANCE CRITERIA'/
$       ' FOR RESIDUAL STRUCTURE A-SET' $
$       MESSAGE // ' ' $
$       IF (EVAL<-2)MATPRN UGVEXT// $
$       UPARTN USET,UGVEXT/TESTA,,,'G'/'A'/'O'/1 $
$
$   CALCULATE MODAL ASSURANCE CRITERIA
$   TRNSP PHIA/PHIAT $
$   MPYAD PHIAT,PHIA,/PHIASQ $
$   TRNSP TESTA/TESTAT $
$   MPYAD TESTAT,TESTA,/TESTASQ $
$
$   Get number of modes
$
$   TYPE PARM,,I,N,NTEST=0,NPHIA=0,IP,IT,TS,TE,ndof $

```

```

TYPE PARM,,RS,N,MAC,PSQ,TSQ,MTEMP $
PARAML TESTA//'TRAILER'/1/S,N,NTEST $
PARAML PHIA//'TRAILER'/1/S,N,NPHIA $
$
MESSAGE //'MODAL ASSURANCE CALCULATION IS PERFORMED FOR '/NPHIA/
      ' CALCULATED MODES AND '/NTEST/' INPUT MODES' $
MESSAGE // ' ' $
MATGEN ,/MACA/7/NPHIA/NTEST $ AD1997
$
$ LOOP ON CALCULATED MODES
$
IP = 1 $
DO WHILE (IP<=NPHIA) $
  TS = IP-1 $
  TE = NPHIA-IP $
  MATGEN ,/PARTP/6/NPHIA/TS/1/TE $
  PARTN PHIA,PARTP,,MODEP,/1 $
  PARAML PHIASQ//'DMI'/IP/IP/S,N,PSQ $
$
$ LOOP ON TEST MODES
$
IT = 1 $
DO WHILE (IT<=NTEST) $
  TS = IT-1 $
  TE = NTEST-IT $
  MATGEN ,/PARTT/6/NTEST/TS/1/TE $
  PARTN TESTA,PARTT,,MODET,/1 $
  MPYAD MODET,MODEP,/MACM/1 $
  PARAML TESTASQ//'DMI'/IT/IT/S,N,TSQ $
  PARAML MACM//'DMI'/1/1/S,N,MTEMP $
  MAC = MTEMP*MTEMP/(PSQ*TSQ) $
TYPE PARM,,CS,N,CMPMAC $ AD1997
$
MESSAGE //' ANALYTICAL MODE '/IP/' COMPARED TO INPUT MODE '/
      IT/ ' - MAC = '/MAC $
      MATGEN ,/PARTIP/4/1/NPHIA/0/1/NPHIA/IP/1/1 $ AD1997
      MATGEN ,/PARTIT/4/1/NTEST/0/1/NTEST/IT/1/1 $ AD1997
      TRNSP PARTIT/PARTITT $ AD1997
      MPYAD PARTIP,PARTITT,/IPITT/ $ AD1997
      CMPMAC=CMPLX(MAC,0.0) $ AD1997
      ADD IPITT,MACA/MACB/CMPMAC//0 $ AD1997
      DELETE /MACA,,,, $ AD1997
      COPY MACB/MACA/ALWAYS $ AD1997
      IT = IT + 1 $
ENDDO $
IP = IP + 1 $
ENDDO $
$
$ add phiT*M*input
$
OUTPUT4 MACA,,,-1/11/-1 $ AD1997
$
$ normalize input modes to unit mass
$
smpyad testa,mmaa,testa,,/tempa/3///1 $
diagonal tempa/scale2/'SQUARE'/2. $ square terms on diagonal
diagonal scale2/scale1//.25 $ sqrt of abs(original value)
PARAML TESTA//'TRAILER'/2/S,N,NDOF $
matgen ,/ident/6/ndof/ndof $ matrix of unit values

```

```

trnsp scale1/scale1t $
mpyad ident,scale1t,/scale $
add testa,scale/nctest///2 $
$
smpyad ntest,mmaa,ntest,,,/checkit/3////1 $
MESSAGE // ' ' $
message //'Normalized check of input modes (after scaling) '/
      'using reduced mass matrix' $
matprn checkit// $
$OUTPUT4 CHECKIT,,,/-1/11/-1 $ AD1997
$
smpyad phia,mmaa,ntest,,,/orthoa/3////1 $
MESSAGE // ' ' $
message //'Results of cross-orthogonality test' $
message //'Each row represents one analytical mode, each column'/
      ' represents an input mode' $
matprn orthoa// $
OUTPUT4 ORTHOA,,,/0/11/-1 $ AD1997
OUTPUT4 PHIA,MMAA,NTEST,/0/13/-1 $ AD1997
$ OUTPUT4 MMAA,PHIA,NTEST,/0/13/-1 $ AD1997
$
$ done
$
ENDIF $ END OF MODAL ASSURANCE CRITERIA CALCULATION
$
$ Write matrices out for use in post-test checking
$ Once again DMIG format is used to minimize the possiblilty of errors
$
matgen eqexins/intext/9/0/lusets $
umerge uset,phia,/phig/'G'/'A'/'O' $
mpyad intext,phig,/phiext/1 $
matmod phiext,eqexins,,,/mat1,/16/1 $
$
umerge uset,testa,/testg/'G'/'A'/'O' $
mpyad intext,testg,/phiex2/1 $
matmod phiex2,eqexins,,,/mat3,/16/1 $
$
umergel uset,mmaa,,,/maag/'G'/'A'/'L' $
mpyad intext,maag,/maal/1 $
mpyad maal,intext,/maaext $
matmod maaext,eqexins,,,/mat2,/16/1 $
MESSAGE // ' ' $
MESSAGE //'PHIEXT (MODES) AND MAAEXT (MASS) OUTPUT AS DMIG' $
LAMX, ,LAMA/LMAT/-1 $
MATPCH LMAT// $
$

```

FIGURE-4: DMAP FOR ORTHO CHECK IN SOL 103 IN VERSION 68.1

```

ALTER 116 $ BEFORE RETURN

OUTPUT4 PHIA,MMAA,,//0/13/-1 $ AD1997
$
$ Write matrices out for use in post-test checking
$ Once again DMIG format is used to minimize the possiblilty of errors
$
matgen eqexins/intext/9/0/lusets $
umerge uset,phia,/phig/'G'/'A'/'O' $
mpyad intext,phig,/phiext/1 $
matmod phiext,eqexins,,//mat1,/16/1 $
$
MESSAGE // ' ' $
MESSAGE //'PHIEXT (MODES) OUTPUT AS DMIG' $
LAMX, ,LAMA/LMAT/-1 $
MATPCH LMAT// $
MESSAGE // ' ' $
MESSAGE //'LAMA (FREQUENCIES) OUTPUT AS PCH' $
$

```

FIGURE-5: DMAP FOR A-SET MODESHAPE IN SOL 103 IN VERSION 68.1

```

COMPILE MODERS SOUIN=MSCSOU, NOREF LIST $

ALTER 103 $

PARAML USET//'TRAILER'/2/S,N,LUSETS $ G-SET SIZE

MATGEN EQEXINS/INTEXT1/9/0/LUSETS $

UPARTN USET,PHIX/PHIX2,,/'A'/'U2'/'Q'/1 $

UMERGE USET,PHIX2,/ZUZR04/'G'/'U2'/'A' $

MPYAD INTEXT1,ZUZR04,/PHIXU2/1 $
mpyad intext1,phixu2,/phiext/1 $
matmod phiext,eqexins,,//mat1,/16/1 $
LAMX, ,LAMA/LMAT/-1 $
MATPCH LMAT// $

```

FIGURE-6: DMAP FOR U-SET MODESHAPE IN SOL 103 IN VERSION 68.1